

The Topsy Hangover: Avoiding Indent Headaches in SAS® Reports

Lisa Mendez, PhD, Army MEDCOM; Richann Jean Watson, DataRich Consulting

ABSTRACT

Creating visually appealing SAS reports shouldn't feel like recovering from a formatting hangover—but when it comes to hanging indents in REPORT procedure, things can get a little... tipsy. In this paper, we serve up a curated cocktail of techniques using a real-world example to help you straighten out those stubborn hanging indents. We'll explore multiple ways to achieve indentation in PROC REPORT using ODS options and clever preprocessing methods. You'll learn why TWIPS (twentieth of a point) matter more than you think, and how understanding them can help you fine-tune your layout with precision. We'll also explain why ODS ESCAPECHAR and the SPLIT option should never share a drink, and how to preprocess variables with line breaks and non-breaking spaces for smoother text flow. As always, we provide a curated list of references to help with other PROC REPORT formatting tips. Whether you're formatting footnotes, crafting multi-line cells, or just trying to avoid the dreaded indent misalignment, this paper offers tips, tricks, and troubleshooting advice to help you sober up your reports and keep your formatting headache-free and avoid the dreaded hangover.

INTRODUCTION

With virtually every study, the presentation of data is essential—much like how a good cocktail needs the right garnish to really shine. SAS offers a variety of tools that let a programmer shake, stir, and serve data visualizations, from crisp charts to smooth, well-balanced reports. PROC REPORT, in particular, is the top-shelf spirit of the bunch: powerful, versatile, and capable of producing vital summaries that go down easy. It's highly customizable, too, though every now and then getting it to behave exactly as you want can feel a bit like trying to read a PROC step after your third margarita—possible, but suddenly more adventurous than expected. Building on that, PROC REPORT is basically a way of turning data into a clean, organized, good-looking report. You provide the data set, and it acts like a very sober bartender who can mix the data into something polished, structured, and easy to digest. You tell it what columns you want, how to group things, what to summarize, and how fancy or simple the final layout should be.

If you haven't yet dabbled with PROC REPORT, or if your memory of it feels a bit blurry, like recalling last night's cocktail lineup, Art Carpenter's *PROC REPORT Basics: Getting Started with the Primary Statements* and Jane Erslinger's *The REPORT Procedure and ODS Destination for Microsoft Excel: The Smarter, Faster Way to Create First-Rate Excel Reports* make excellent refreshers. And if you've already crafted a few basic reports, then this paper arrives with perfect timing to help you dodge the dreaded hanging indent and keep you stepping forward with the confidence of someone who's only slightly tipsy.

SETTING THE BAR

When you sashay up to the bar, you need to have in mind what kind of cocktail you want. Otherwise, you leave the bartender to their own imaginations, and you never know what you are going to end up with. Similarly, when producing reports, you need to have a general idea of what the end goal is. As a bartender is equipped with the right tools to craft that perfect cocktail, you need to make sure you have all the right tools at your disposal to create an aesthetically pleasing report.

DESIRED OUTPUT

You can reference the Statistical Analysis Plan (SAP) or mock shells, to know what the end goal is. You need to produce the Sample Listing 1. Although at first glance it seems straightforward and like sauntering up to the bar with confidence and ordering exactly what you want, you can say "I can produce that". However, after further examination you notice that you need to indent the preferred term and allow for a hanging indent. You then need to further indent the verbatim term and allow for a hanging indent at that indentation level. In addition, the footnote needs to have indentation. You also notice that there is special styling on the third column that requires bold and italics. Your simple report has now become complex, which can be a recipe for disaster in the hands of someone less experienced.

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA APPLICATION SITE ERYTHEMA	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS APPLICATION SITE PRURITUS	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA DIARRHOEA	09JAN2014 (8)	11JAN2014 (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
 If the AE start date only contains the month and year and has the same month and year as the exposure start date,
 then it is considered treatment emergent.
 If the AE start date is only the year and the year is the same as the exposure start date, then it is considered
 treatment emergent.

Sample Listing 1: Desired Output Provided from Table, Listing and Figure Mock Shells

SAMPLE DATA

After you have selected the correct mock shell (or recipe if you are looking to make a cocktail), you need to gather all the ingredients. In the case of the report, our main ingredient is the data. Sample Data 1 is our plain data set, nothing fancy and can be used as-is if the report was simple.

Sample Data 1 contains the first ten records from the CDISC pilot study ADAE data set (some variables removed due to space).

Row	USUBJID	ASTDT	AETERM	AEDECOD
1	01-701-1015	03JAN2014	APPLICATION SITE ERYTHEMA	APPLICATION SITE ERYTHEMA
2	01-701-1015	03JAN2014	APPLICATION SITE PRURITUS	APPLICATION SITE PRURITUS
3	01-701-1015	09JAN2014	DIARRHOEA	DIARRHOEA
4	01-701-1023	07AUG2012	ERYTHEMA	ERYTHEMA
5	01-701-1023	07AUG2012	ERYTHEMA	ERYTHEMA
6	01-701-1023	07AUG2012	ERYTHEMA	ERYTHEMA
7	01-701-1023	26AUG2012	ATRIOVENTRICULAR BLOCK SECOND DEGREE	ATRIOVENTRICULAR BLOCK SECOND DEGREE
8	01-701-1028	21JUL2013	APPLICATION SITE ERYTHEMA	APPLICATION SITE ERYTHEMA
9	01-701-1028	08AUG2013	APPLICATION SITE PRURITUS	APPLICATION SITE PRURITUS
10	01-701-1034	27AUG2014	APPLICATION SITE PRURITUS	APPLICATION SITE PRURITUS

Row	AEBODSYS	AESEV	AEREL
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS	MILD	PROBABLE
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS	MILD	PROBABLE
3	GASTROINTESTINAL DISORDERS	MILD	REMOTE
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS	MILD	POSSIBLE
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS	MODERATE	PROBABLE
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS	MILD	POSSIBLE
7	CARDIAC DISORDERS	MILD	POSSIBLE
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS	MILD	POSSIBLE
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS	MILD	PROBABLE
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS	MILD	PROBABLE

Row	AEOUT	TRTEMFL	AOCCSFL	AOCCPFL
1	NOT RECOVERED/NOT RESOLVED	Y	Y	Y
2	NOT RECOVERED/NOT RESOLVED	Y		Y
3	RECOVERED/RESOLVED	Y	Y	Y
4	NOT RECOVERED/NOT RESOLVED	Y	Y	Y
5	NOT RECOVERED/NOT RESOLVED	Y		
6	RECOVERED/RESOLVED	Y		
7	NOT RECOVERED/NOT RESOLVED	Y	Y	Y
8	NOT RECOVERED/NOT RESOLVED	Y	Y	Y
9	NOT RECOVERED/NOT RESOLVED	Y		Y
10	NOT RECOVERED/NOT RESOLVED	Y	Y	Y

Sample Data 1: Adverse Event

PREPPING THE DATA

Before getting started a good bartender will have prepped any necessary ingredients to speed up the process, thus, you examine the main ingredient (Sample Data 1) and the recipe (Sample Listing 1) and notice that due to the amount of information on the listing and the limitations with producing an output with 'Letter' paper size allowing for a one-inch margin, the data needs to be prepped by combining some columns (Sample Data 2).

Row	SOC_PT_TERM
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE ERYTHEMA/APPLICATION SITE ERYTHEMA
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE PRURITUS/APPLICATION SITE PRURITUS
3	GASTROINTESTINAL DISORDERS/DIARRHOEA/DIARRHOEA
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEMA
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEMA
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEMA
7	CARDIAC DISORDERS/ATRIOVENTRICULAR BLOCK SECOND DEGREE/ATRIOVENTRICULAR BLOCK SECOND DEGREE
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE ERYTHEMA/APPLICATION SITE ERYTHEMA
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE PRURITUS/APPLICATION SITE PRURITUS
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE PRURITUS/APPLICATION SITE PRURITUS

Row	SAR	STDT_DY	ENDT_DY
1	F / 63 / W	03JAN2014 / (2)	ONGOING
2	F / 63 / W	03JAN2014 / (2)	ONGOING
3	F / 63 / W	09JAN2014 / (8)	11JAN2014 / (10)
4	M / 64 / W	07AUG2012 / (3)	30AUG2012 / (26)
5	M / 64 / W	07AUG2012 / (3)	ONGOING
6	M / 64 / W	07AUG2012 / (3)	30AUG2012 / (26)
7	M / 64 / W	26AUG2012 / (22)	ONGOING
8	M / 71 / W	21JUL2013 / (3)	ONGOING
9	M / 71 / W	08AUG2013 / (21)	ONGOING
10	F / 77 / W	27AUG2014 / (58)	ONGOING

Sample Data 2: Adverse Event with Fields Combined for Listing

PRODUCING THE OUTPUT

Like the bartender who has their recipe and the ingredients and has done all the necessary prep work, you are ready to produce the report. You craft a beautiful program (SAS Program 1) that uses Sample Data 2 to produce Output 1. However, there is no indentation for the combined field of System Organ Class / Preferred Term / Verbatim Term. This is not what you wanted, and you need to go back and either revisit the ingredients or the prep work.

```

footnote j = 1 'Adverse event is defined as treatment emergent if it occurred on or after the
start of exposure to study drug. If the AE start date only contains the month and year and has
the same month and year as the exposure start date, then it is considered treatment emergent. If
the AE start date is only the year and the year is the same as the exposure start date, then it
is considered treatment emergent.';

ods escapechar = '`';

proc report data = ae split = '~'
    style(report) = [rules = none frame = void]
    style(header) = [background = white borderbottomwidth = 2 borderbottomcolor = black];
by TRTA;
columns USUBJID SAR SOC_PT_TERM RTF STDT_DY ENDT_DY AESEV AEREL AEOUT;
define USUBJID / order order = data 'Subject'
    style(header) = [just = left] style(column) = [cellwidth = 9%];
define SAR / group 'Sex / Age / Race' style(column) = [cellwidth = 8%];
define SOC_PT_TERM / display left "System Organ Class/~Preferred Term/~Verbatim Term"
    style(header) = [just = left] style(column) = [cellwidth = 23%];
define STDT_DY / display 'Start Date (Day)' style(column) = [cellwidth = 10%];
define ENDT_DY / display 'End Date (Day)' style(column) = [cellwidth = 10%];
define AESEV / display 'Severity' style(column) = [cellwidth = 10%];
define AEREL / display left 'Relation to~Study Drug'
    style(column) = [cellwidth = 10%];
define AEOUT / display left 'Outcome' style(column) = [cellwidth = 19%];

compute after _page_ / style = {bordertopcolor = black bordertopwidth = 2};
line ' ';
endcomp;
run;

```

SAS Program 1: PROC REPORT to Produce Listing

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE ERYTHEMA/APPLICATION SITE ERYTHEMA	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/APPLICATION SITE PRURITUS/APPLICATION SITE PRURITUS	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/DIARRHOEA/DIARRHO EA	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEM A	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEM A	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ERYTHEMA/ERYTHEM A	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ATRIOVENTRICULAR BLOCK SECOND DEGREE/ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

Output 1: Listing of All Adverse Events - No Indentation without Styling

RTF OUTPUT DESTINATION

THE DREADED HANGOVER

As you examine your beautiful masterpiece with heartache, you think of where you can adjust to achieve the desired output. You start with something that may be easier to adjust, like substituting a lemon for a lime in a margarita (i.e., the styling of column 3) and leave the harder stuff to the end like making your own flavored simple syrup (i.e., the hanging indentations).

First things first, you need to go back to the prep station. To account for the need for bold and italics, the style (red font) is also embedded in the combined field. In addition, to ensure that preferred term and verbatim term start on a new row, the following RFT sequence, `n` (green font), is incorporated into the combined field. The grave accent (`) is used as the escape character (it's a signal to tell RTF to do something, like waving to get the bartender's attention). Refer to Sample Data 3 for an illustration of the newly prepped data.

Row	SOC_PT_TERM
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i>
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>
3	GASTROINTESTINAL DISORDERS/ DIARRHOEA / <i>DIARRHOEA</i>
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
7	CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE / <i>ATRIOVENTRICULAR BLOCK SECOND DEGREE</i>
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i>
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>

Row	SAR	STDT_DY	ENDT_DY
1	F / 63 / W	03JAN2014 / (2)	ONGOING
2	F / 63 / W	03JAN2014 / (2)	ONGOING
3	F / 63 / W	09JAN2014 / (8)	11JAN2014 / (10)
4	M / 64 / W	07AUG2012 / (3)	30AUG2012 / (26)
5	M / 64 / W	07AUG2012 / (3)	ONGOING
6	M / 64 / W	07AUG2012 / (3)	30AUG2012 / (26)
7	M / 64 / W	26AUG2012 / (22)	ONGOING
8	M / 71 / W	21JUL2013 / (3)	ONGOING
9	M / 71 / W	08AUG2013 / (21)	ONGOING
10	F / 77 / W	27AUG2014 / (58)	ONGOING

Sample Data 3: Adverse Event with Fields Combined for Listing with Styling

SAS Program 1 using Sample Data 3 produces the output with the bold and italics styling as seen in RTF Output 1. Ah! The simple joys. You can now enjoy a sip of that cocktail. You are on the right track. Now to add the hanging indents.

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ <i>APPLICATION SITE ERYTHEMA</i>	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ <i>APPLICATION SITE PRURITUS</i>	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA/ <i>DIARRHOEA</i>	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ <i>ERYTHEMA</i>	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ <i>ERYTHEMA</i>	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ <i>ERYTHEMA</i>	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ <i>ATRIOVENTRICULAR BLOCK SECOND DEGREE</i>	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

RTF Output 1: Listing of All Adverse Events – No Indentation with Styling

Although the cocktail is good, it is not great. You need to do some minor tweaks. You need to get those hanging indents added. So back to the prep station. This time you embed blank spaces to Sample Data 3 to force an indent (Sample Data 4 **Error! Reference source not found.**).

Row	SOC_PT_TERM
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ ERYTHEMA}
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ PRURITUS}
3	GASTROINTESTINAL DISORDERS/ DIARRHOEA/ DIARRHOEA}
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA}
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA}
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA}
7	CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE}
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ ERYTHEMA}
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ PRURITUS}
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ PRURITUS}

Sample Data 4: Adverse Event with Fields Combined for Listing with Styling and Embedded Spaces

Using a slightly revised recipe, you adjust your masterpiece and add some embedded spaces in the labels and add a new option (ASIS = OFF) to the header and column style (SAS Program 2). This ends up producing the exact same output as RTF Output 1, in other words, the spacing was not preserved.

```
ods escapechar = '~';
proc report data = ae split = '~'
  style(report) = [rules = none frame = void]
  style(header) = [background = white borderbottomwidth = 2 borderbottomcolor = black];
  by TRTA;
  columns USUBJID SAR SOC_PT_TERM RTF STDT_DY ENDT_DY AESEV AEREL AEOUT;
  define USUBJID / order order = data 'Subject'
    style(header) = [just = left] style(column) = [cellwidth = 9%];
  define SAR / group 'Sex / Age / Race' style(column) = [cellwidth = 8%];
  define SOC_PT_TERM / display left
    "System Organ Class/~ Preferred Term/~ Verbatim Term"
    style(header) = [just = left ASIS = OFF]
    style(column) = [cellwidth = 23% ASIS = OFF];
  define STDT_DY / display 'Start Date (Day)' style(column) = [cellwidth = 10%];
  define ENDT_DY / display 'End Date (Day)' style(column) = [cellwidth = 10%];
  define AESEV / display 'Severity' style(column) = [cellwidth = 10%];
  define AEREL / display left 'Relation to~Study Drug'
    style(column) = [cellwidth = 10%];
  define AEOUT / display left 'Outcome' style(column) = [cellwidth = 19%];

  compute after _page_ / style = {bordertopcolor = black bordertopwidth = 2};
  line ' ';
endcomp;
run;
```

SAS Program 2: PROC REPORT to Produce Listing with ASIS = OFF

Using SAS Program 2, let's try setting ASIS = ON. Now this is much closer to what we were looking for (RTF Output 2). However, notice the portion highlighted in pink. It indented the first line of the indicated portion of the text, but when that piece of the text was wrapped to the next line, it began at the beginning of the line.

Note that ASIS by default is OFF. ASIS = ON indicates that embedded spacing and line breaks should be preserved.

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ APPLICATION SITE ERYTHEMA	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ APPLICATION SITE PRURITUS	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA/ DIARRHOEA	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

RTF Output 2: Listing of All Adverse Events – No Indentation – Same as SAS Program 2 with ASIS = ON

Every bartender has their own spin on things and does things slightly differently; programmers are no different. You may want to use the `n` with the embedded spaces techniques (**Error! Reference source not found.**and SAS Program 2) while others may want to explore alternative ways.

One technique is to use RTF markup for tabs as shown in Sample Data 5 and SAS Program 3. In SAS Program 3 only the DEFINE statement for the third column is shown since the rest of the PROC REPORT is the same. This produces the same output as RTF Output 2.

- **\dftab180** indicates the tab size. RTF measurements are normally in TWIPs (twentieth of a point). 180 TWIPs is approximately 0.125 inches.
- **\tqr** indicates that the tab is right aligned
- **\tab** indicates to insert a tab
(Rich Text Format (RTF) Version 1.5 Specification, n.d.)

Whichever method you use is your preference, just like if you're in the mood for an espresso martini, or a chocolate beer with peanut butter whisky.

Row	SOC_PT_TERM
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i>
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>
3	GASTROINTESTINAL DISORDERS/ DIARRHOEA / <i>DIARRHOEA</i>
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i>
7	CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE / <i>ATRIOVENTRICULAR BLOCK SECOND DEGREE</i>
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i>
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i>

Sample Data 5: Adverse Event with Fields Combined for Listing with Styling and RTF Tab Markup

```

...
define SOC_PT_TERM / display left "System Organ Class/Preferred
Term/Verbatim Term"
style(header) = [just = left]
style(column) = [cellwidth = 23%];
...

```

SAS Program 3: DEFINE Statement with RTF Tab Markup

ALL INDENTS AT THE SAME LEVEL

Using various techniques in RTF did not allow for multiple hanging indents, but we were able to achieve a single-level hanging indentation. Updating SAS Program 2 with the changes noted in SAS Program 4, you have crafted something that is tolerable, like an espresso martini with coffee liqueur instead of real coffee.

As mentioned previously, ASIS = ON allows for the preservation of any embedded spaces or line breaks but does not mask any RTF markup. PROTECTSPECIALCHARS by default is ON and this protects the characters so that they are not interpreted as markup language. RTF Output 3 demonstrates what happens if PROTECTSPECIALCHARS = ON. Setting PROTECTSPECIALCHARS = OFF allows the characters to be treated as markup language if necessary. PRETEXT is another option that has been added and this allows you to place text before the table cell. The PRETEXT is RTF markup.

- **\pnhang** indicates that the cell is going to use a hanging indent
- **\fi-220** indicates that the first line should be set to -220 TWIPs, which moves it to the left
- **\li220** indicates that the entire paragraph should be moved 220 TWIPs to the right (Rich Text Format (RTF) Version 1.5 Specification, n.d.)

These changes allow for a single-level indentation, but it preserves the indentation on the wrapped rows (RTF Output 4).

```

...
define SOC_PT_TERM / display left
  "System Organ Class/~ Preferred Term/~ Verbatim Term"
style(header) = [just = left ASIS = on]
style(column) = [cellwidth = 23% PROTECTSPECIALCHARS = off
  pretext = '\pnhang\fi-220\li220'];
...

```

SAS Program 4: DEFINE Statement with PROTECTSPECIALCHARS = OFF and PRETEXT Option

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	\pnhang\fi-220\li220GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ APPLICATION SITE ERYTHEMA	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		\pnhang\fi-220\li220GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ APPLICATION SITE PRURITUS	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		\pnhang\fi-220\li220GASTROINTESTINAL DISORDERS/ DIARRHOEA/ DIARRHOEA	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	\pnhang\fi-220\li220SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		\pnhang\fi-220\li220SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		\pnhang\fi-220\li220SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		\pnhang\fi-220\li220CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered

RTF Output 3: Listing of All Adverse Events – Indentation – ASIS = ON and PROTECTSPECIALCHARS = ON

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ APPLICATION SITE ERYTHEMA	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ APPLICATION SITE PRURITUS	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA/ DIARRHOEA	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

RTF Output 4: Listing of All Adverse Events – Indentation – ASIS = ON and PROTECTSPECIALCHARS = OFF

RTF FOOTNOTES

Wow! That was a bit tedious. Time to taste our concoction and see if we like it. Ah! There is still something missing. The footnotes need to be indented as well. Just like a pina colada needs a slice of pineapple for garnish.

In order to get the indented footnote, let's try some similar techniques that we tried for indenting the output. Let's modify SAS Program 2 with ASIS = ON to have the following footnote and add the embedded spaces and the line break.

```
footnote j = 1 'Adverse event is defined as treatment emergent if it occurred on or after the
start of exposure to study drug. `n` If the AE start date only contains the month and year
and has the same month and year as the exposure start date, then it is considered treatment
emergent. `n` If the AE start date is only the year and the year is the same as the exposure
start date, then it is considered treatment emergent.';
```

SAS Program 5: FOOTNOTE Statement with Embedded Spaces

Well, that was a no-go. It didn't preserve the spaces in the footnote. The ASIS = ON option only applies to the data in the report not the footnote statements. Let's try the tab markup similar to the modification for the output in SAS Program 3.

```
footnote j = 1 "Adverse event is defined as treatment emergent if it occurred on or after the
start of exposure to study drug. `n`R`\deftab180\tqr\tab\tab If the AE start date only contains
the month and year and has the same month and year as the exposure start date, then it is
considered treatment emergent.`n`R`\deftab180\tqr\tab\tab If the AE start date is only the
year and the year is the same as the exposure start date, then it is considered treatment
emergent.>";
```

SAS Program 6: FOOTNOTE Statement with RTF Tab Markup

Using the footnote in SAS Program 6 gives us the indentation but when it wraps it starts back at the beginning (RTF Output 5).

```
Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
  If the AE start date only contains the month and year and has the same month and year as the exposure start date, then
it is considered treatment emergent.
  If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment
emergent.
```

RTF Output 5: Footnote with RTF Tab Markup

So close...but we are missing one final ingredient. We need that footnote to be a hanging indent. Let's try the technique similar to what is in SAS Program 4. We append the hanging indent markup to the front of the footnote.

```
footnote j = 1 "`R/RTF`\pnhang\fi-220\li220' Adverse event is defined as treatment emergent if
it occurred on or after the start of exposure to study drug. If the AE start date only contains
the month and year and has the same month and year as the exposure start date, then it is
considered treatment emergent. If the AE start date is only the year and the year is the same
as the exposure start date, then it is considered treatment emergent.>";
```

SAS Program 7: FOOTNOTE Statement with RTF Hanging Indent Markup

SAS Program 7 gives us the following footnotes shown in RTF Output 6 which is what we were looking for but the problem is that to achieve this the option BODYTITLE must be used on the ODS RTF statement. If the option NOBODYTITLE is used, then the indentation is not preserved.

```
Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
  If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it
is considered treatment emergent.
  If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment
emergent.
```

RTF Output 6: Footnote with RTF Hanging Indent Markup

The biggest hurdle is finding a technique that will allow for indentation within the output as well as the footnotes regardless of whether BODYTITLE or NOBODYTITLE is used.

For ODS RTF destination the best combination to achieve close to the desired goal is using the hanging indent technique in SAS Program 4 and moving the footnote out of the footnote statement into a compute block. As with any great concoction, you need to save that before you forget, so we are making note of the full code (SAS Program 8) that yielded the optimal results (for ODS RTF). RTF Output 7 is pretty close to what we originally wanted and maybe we can live with it. After all there are some compromises that may be needed when crafting with items that may have limitations. Like crafting a cocktail when you don't have the right ingredients. You make substitutions and enjoy it anyway.

A note about using LINE statements in the compute block: the placement of the footnote may be different on each page because it is not a true page-level footnote. It is just another row in the report output, so its position depends on where the report happens to end on that page, which can shift.

```

ods escapechar = ' ';
proc report data = ae split = '~'
  style(report) = [rules = none frame = void]
  style(header) = [background = white borderbottomwidth = 2 borderbottomcolor = black];
  by TRTA;
  columns USUBJID SAR SOC_PT_TERM RTF STDT_DY ENDT_DY AESEV AEREL AEOUT;
  define USUBJID / order order = data 'Subject'
  style(header) = [just = left] style(column) = [cellwidth = 9%];
  define SAR / group 'Sex / Age / Race' style(column) = [cellwidth = 8%];
  define SOC_PT_TERM / display left
  "System Organ Class/~ Preferred Term/~ Verbatim Term"
  style(header) = [just = left ASIS = OFF]
  style(column) = [cellwidth = 23%
  PROTECTSPECIALCHARS = off
  pretext = '\pnhang\fi-220\li220'];
  define STDT_DY / display 'Start Date (Day)' style(column) = [cellwidth = 10%];
  define ENDT_DY / display 'End Date (Day)' style(column) = [cellwidth = 10%];
  define AESEV / display 'Severity' style(column) = [cellwidth = 10%];
  define AEREL / display left 'Relation to Study Drug'
  style(column) = [cellwidth = 10%];
  define AEOUT / display left 'Outcome' style(column) = [cellwidth = 19%];

  compute after _page_ / style = {bordertopcolor = black bordertopwidth = 2
  ASIS = ON JUST = L};

  line ' ';
  line 'Adverse event is defined as treatment emergent if it occurred on or after the start
of exposure to study drug.';
  line ' If the AE start date only contains the month and year and has the same month and
year as the exposure start date, then it is considered treatment emergent.';
  line ' If the AE start date is only the year and the year is the same as the exposure
start date, then it is considered treatment emergent.';
endcomp;
run;

```

SAS Program 8: Final Version of RTF with Single-Level Hanging Indents for Data and Footnote

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA/ APPLICATION SITE ERYTHEMA	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS/ APPLICATION SITE PRURITUS	03JAN2014 / (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA/ DIARRHOEA	09JAN2014 / (8)	11JAN2014 / (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA/ ERYTHEMA	07AUG2012 / (3)	30AUG2012 / (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 / (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent.
If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

RTF Output 7: Final RTF with Single-Level Hanging Indents for Data and Footnote

Something to point out is that you need to make sure that the ODS ESCAPECHAR and the SPLIT character on the PROC REPORT statement are not the same. PROC REPORT split character is applied before the raw RTF is rendered. Thus, due to the timing, the raw RTF command is no longer a complete command thus leaving an undesired outcome. Thanks to our Bailout Buddy, Jane Eslinger, for giving us that important tip, and holding us up at the bar when we start to get tipsy.

PDF OUTPUT DESTINATION

OVERCOME THE DREADED HANGOVER

Similar to swapping out rare ingredients with common but similar ingredients in that cocktail that you just ordered, you swap out for a different ODS destination to see if it will achieve the desired result. We demonstrated how to indent and create a hanging indent with RTF, now we turn to replicating the same technique with the ODS PDF destination. Using Sample Data 3 and SAS Program 2 and switching to ODS PDF destination, it yields a similar output as RTF Output 1 (see PDF Output 1).

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA APPLICATION SITE ERYTHEMA	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS APPLICATION SITE PRURITUS	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA DIARRHOEA	09JAN2014 (8)	11JAN2014 (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

PDF Output 1: Listing of All Adverse Events – No Indentation – ASIS = OFF

On the DEFINE statement in PROC REPORT, if we use ASIS = ON instead of ASIS = OFF in SAS Program 2, only the column headers are indented because the spacing is preserved and there was no additional spacing added in combined field in the data set. However, if we use the modified combined field that includes the embedded spacing as seen in Sample Data 4 **Error! Reference source not found.**, then using ASIS = ON in SAS Program 2 preserves the spaces in the data values as well. However, as seen in PDF Output 2 when the preferred term and/or verbatim term wrap to the next line, it is left justified and is not indented to the same level (i.e., a hanging indent). By now you are thinking, I need that cocktail. Like a simple drink that has lots of modifications, a simple report can turn into a disaster when a

bunch of customization is added.

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA APPLICATION SITE ERYTHEMA	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS APPLICATION SITE PRURITUS	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA DIARRHOEA	09JAN2014 (8)	11JAN2014 (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE/ ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

PDF Output 2: Listing of All Adverse Events - Indentation (No Hanging) – ASIS = ON

SUCCESSFUL HANGING INDENTS

We saw that with RTF we were able to do a hanging indent but only at one level and for the most part this would be adequate. But in order to achieve the multi-level hanging indentation, we need to use PDF destination, and we need to introduce a marker with the escape character (`m) into our combined field. With this escape sequence (`m), it indicates that this is where the text should start when it wraps. Thus, we can set an escape sequence for the preferred term to have one level of indentation and a second escape sequence for the verbatim term to have a second level of indentation. Refer to Sample Data 6 for an illustration of how the escape sequence is implemented.

Row	SOC_PT_TERM
1	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i> }
2	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i> }
3	GASTROINTESTINAL DISORDERS/ DIARRHOEA / <i>DIARRHOEA</i> }
4	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i> }
5	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i> }
6	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA / <i>ERYTHEMA</i> }
7	CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE / <i>ATRIOVENTRICULAR BLOCK SECOND DEGREE</i> }
8	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA / <i>APPLICATION SITE ERYTHEMA</i> }
9	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i> }
10	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS / <i>APPLICATION SITE PRURITUS</i> }

Sample Data 6: Adverse Event with Fields Combined for Listing with Styling and Embedded Spaces and Marker for Wrapping

With the escape sequence used to mark where the values should wrap to, SAS Program 2 modified to use ASIS = ON can be re-run and it yields PDF Output 3 which now shows the hanging indent for both preferred term and verbatim term.

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA APPLICATION SITE ERYTHEMA	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS APPLICATION SITE PRURITUS	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA DIARRHOEA	09JAN2014 (8)	11JAN2014 (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA ERYTHEMA	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE ATRIOVENTRICULAR BLOCK SECOND DEGREE	26AUG2012 (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.

PDF Output 3: Listing of All Adverse Events - Indentation (Hanging) - ASIS = ON

An alternative to SAS Program 2 modified to use ASIS = ON is SAS Program 9. In SAS Program 9, instead of specifying ASIS = ON as an option on the DEFINE statements this can be specified on the PROC REPORT statement in the STYLE(HEADER) and STYLE(COLUMN) options. Specifying it on the PROC REPORT statement, indicates that ASIS = ON applies to all headers and all columns in the report.

```
proc report data = ae split = '~'
  style(report) = [rules = none frame = void]
  style(header) = [background = white borderbottomwidth = 2
    borderbottomcolor = black ASIS = on]
  style(column) = [ASIS = on];
by TRTA;
columns USUBJID sar soc_pt_term stdt_dy endt_dy AESEV AEREL AEOUT;
define USUBJID / order order = data 'Subject'
  style(header) = [just = left] style(column) = [cellwidth = 9%];
define sar / group 'Sex / Age / Race' style(column) = [cellwidth = 8%];
define SOC_PT_TERM / display left
  "System Organ Class/~ Preferred Term/~ Verbatim Term"
  style(header) = [just = left] style(column) = [cellwidth = 23%];
define stdt_dy / display 'Start Date~(Day)' style(column) = [cellwidth = 10%];
define endt_dy / display 'End Date~(Day)' style(column) = [cellwidth = 10%];
define AESEV / display 'Severity' style(column) = [cellwidth = 10%];
define AEREL / display left 'Relation to~Study Drug'
  style(column) = [cellwidth = 10%];
define AEOUT / display left 'Outcome' style(column) = [cellwidth = 19%];
compute after _page_ / style = {bordertopcolor = black bordertopwidth = 2};
  line ' ';
endcomp;
run;
```

SAS Program 9: PROC REPORT to Produce Listing with ASIS = ON in PROC REPORT Statement

PDF FOOTNOTES

We are now mastering our mixology. Switching that one ingredient allowed us to achieve the multi-level hanging indent. But we still have the footnotes to contend with. In the mock shell, we notice that the definition for treatment emergent is indented under the main footnote and that there is a hanging indent (i.e., the indentation for subsequent rows for that text string should wrap to the same level). We can try implementing the STYLE with ASIS = ON (``S = {ASIS = ON}`) and embed the spaces similar to what was done with the combined field for the system organ class, preferred term and verbatim term (SAS Program 10).

```
footnote j = 1 "`S = {ASIS = on} Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. `n`m If the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. `n`m If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.";
```

SAS Program 10: Footnote Statement for Listing Using Style with ASIS = ON

Notice that in PDF Output 4 that the second and third footnote did indent but when it wrapped the wrapped line was left justified and did not indent.

```
Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
  If the AE start date only contains the month and year and has the same month and year as the exposure start date, then
  it is considered treatment emergent.
  If the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment
  emergent.
```

PDF Output 4: Footnote Using STYLE ASIS = ON with Escape Sequence for a New Line and Embedded Spaces

To achieve the hanging indent in the footnote, we need to use the marker escape sequence (``m`) that was used in the combined field for the report along with `-n` escape sequence (``-n`). When setting a marked location, the newline (``n`) does not allow a line break to start at the marked location, only ``-n` allows starting a new line at the marked location (Lund, You Did That Report in SAS@!:: The Power of the ODS PDF Destination, 2011). Notice that with the ``m` and ``-n` technique there is no need to add embedded spaces.

```
footnote j = 1 "Adve`mrse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug. `n`-nIf the AE start date only contains the month and year and has the same month and year as the exposure start date, then it is considered treatment emergent. `n`-nIf the AE start date is only the year and the year is the same as the exposure start date, then it is considered treatment emergent.";
```

SAS Program 11: Footnote Statement for Listing Using Marker Sequence

```
Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
  If the AE start date only contains the month and year and has the same month and year as the exposure start date,
  then it is considered treatment emergent.
  If the AE start date is only the year and the year is the same as the exposure start date, then it is considered
  treatment emergent.
```

PDF Output 5: Footnote Using Marker Escape Sequence (`m) with New Line Indicator (`-n)

By combining those perfect ingredients and techniques (SAS Program 9 and SAS Program 11), we have yielded a masterpiece (PDF Output 6), exactly like an Espresso martini with real espresso coffee and vanilla vodka, and three coffee beans for garnish. ¡Delicioso!

Placebo

Subject	Sex / Age / Race	System Organ Class/ Preferred Term/ Verbatim Term	Start Date (Day)	End Date (Day)	Severity	Relation to Study Drug	Outcome
01-701-1015	F / 63 / W	GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE ERYTHEMA <i>APPLICATION SITE ERYTHEMA</i>	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GENERAL DISORDERS AND ADMINISTRATION SITE CONDITIONS/ APPLICATION SITE PRURITUS <i>APPLICATION SITE PRURITUS</i>	03JAN2014 (2)	ONGOING	MILD	PROBABLE	NOT RECOVERED/NOT RESOLVED
		GASTROINTESTINAL DISORDERS/ DIARRHOEA <i>DIARRHOEA</i>	09JAN2014 (8)	11JAN2014 (10)	MILD	REMOTE	RECOVERED/RESOLVED
01-701-1023	M / 64 / W	SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA <i>ERYTHEMA</i>	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA <i>ERYTHEMA</i>	07AUG2012 (3)	ONGOING	MODERATE	PROBABLE	NOT RECOVERED/NOT RESOLVED
		SKIN AND SUBCUTANEOUS TISSUE DISORDERS/ ERYTHEMA <i>ERYTHEMA</i>	07AUG2012 (3)	30AUG2012 (26)	MILD	POSSIBLE	RECOVERED/RESOLVED
		CARDIAC DISORDERS/ ATRIOVENTRICULAR BLOCK SECOND DEGREE <i>ATRIOVENTRICULAR BLOCK SECOND DEGREE</i>	26AUG2012 (22)	ONGOING	MILD	POSSIBLE	NOT RECOVERED/NOT RESOLVED

Adverse event is defined as treatment emergent if it occurred on or after the start of exposure to study drug.
 If the AE start date only contains the month and year and has the same month and year as the exposure start date,
 then it is considered treatment emergent.
 If the AE start date is only the year and the year is the same as the exposure start date, then it is considered
 treatment emergent.

PDF Output 6: Final PDF Output with Multi-Level Hanging Indent in Data and Hanging Indent in Footnote

CONCLUSION

In the world of SAS reports, hanging indents can make the difference between a report that is polished and one that leaves the reader with a bit of a formatting hangover. Much like becoming a mixologist, achieving clean and consistent indentation requires the right balance of ingredients, prep work, tools, and techniques, along with a willingness to experiment.

Through thoughtful processing and a deeper understanding of some useful techniques, we have provided you with some tips to tackle the most stubborn formatting challenges. Although you may have a preference once you begin to master either RTF or PDF, understanding how each one deals with indents allows you flexibility when dealing with different client needs and wants. Each method adds another ingredient to your reporting toolkit, like a bartender's top-shelf selection. While the process may take a bit of trial and error, the end result is worth it.

REFERENCES

- Carpenter, A. (2007). *Carpenter's Complete Guide to the SAS® REPORT Procedure*. Cary, NC: SAS Institute Inc.
- Carpenter, A. (2012). PROC REPORT Basics: Getting Started with the Primary Statements. Orlando, FL: SAS Global Forum. Retrieved from <https://support.sas.com/resources/papers/proceedings12/242-2012.pdf>
- Coar, B. (2016). Indenting with Style. WUSS. Retrieved Nov 2025, from https://www.lexjansen.com/wuss/2016/53_Final_Paper_PDF.pdf
- Eslinger, J. (2016). *The SAS® Programmer's PROC REPORT Handbook: Basic to Advanced Reporting Techniques*. Cary, NC: SAS Institute Inc.
- Eslinger, J. (2017). The REPORT Procedure and ODS Destination for Microsoft Excel: The Smarter, Faster Way to Create First-Rate Excel Reports. Orlando, FL: SAS Global Forum. Retrieved from <https://support.sas.com/resources/papers/proceedings17/SAS0235-2017.pdf>
- Eslinger, J. (2018). *The SAS® Programmer's PROC REPORT Handbook: ODS Companion*. Cary, NC: SAS Institute Inc.
- Eslinger, J. (n.d.). An In-depth Look at PROC REPORT. WILSU. Retrieved Nov 2025, from <https://www.wilsu.org/sdajgfuirHIUTIsdfnloa312/SUSJun2016/Proceedings/Slides/Eslinger%20-%20An%20In-depth%20Look%20at%20PROC%20REPORT.pdf>
- Hamilton, P. (n.d.). ODS to RTF: Tips and Tricks. SUGI. Retrieved Nov 2025, from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi28/024-28.pdf>
- Lund, P. (2009). PDF Can be Pretty Darn Fancy - Tips and Tricks for the ODS PDF Destination. PharmaSUG. Retrieved Nov 2025, from <https://www.lexjansen.com/pharmasug/2009/tu/TU08.pdf>
- Lund, P. (2011). You Did That Report in SAS@!:: The Power of the ODS PDF Destination. SAS Global Forum. Retrieved Jan 2026, from <https://support.sas.com/resources/papers/proceedings11/247-2011.pdf>
- Matthews, C., & Kalchenko, E. (2013). Pretty Please?! Making RTF Output “Pretty” with SAS. PharmaSUG. Retrieved Jan 2026, from <https://pharmasug.org/proceedings/2013/IB/PharmaSUG-2013-IB08.pdf>
- Parsons, L. S. (2006). Enhancing RTF Output with RTF Control Words and In-Line Formatting. PharmaSUG. Retrieved Nov 2025, from <https://www.lexjansen.com/pharmasug/2006/Posters/PO04.pdf>
- Rich Text Format (RTF) Version 1.5 Specification*. (n.d.). Retrieved 11 2025, from Bibloscape 9 Research Information Manager: https://www.bibloscape.com/rtf15_spec.htm
- SAS Institute Inc. (n.d.). *Samples & SAS Notes*. Retrieved Jan 2026, from SAS Support: <https://support.sas.com/kb/42/377.html>
- Srivastva, S. (2017). Text Wrapping with Indentation for RTF Reports. WUSS. Retrieved Nov 2025, from https://www.lexjansen.com/wuss/2017/53_Final_Paper_PDF.pdf
- Zender, C. L. (2010). Proc REPORT Tutorial. WUSS. Retrieved Nov 2025, from https://www.lexjansen.com/wuss/2010/TUT/3027_1_TUT-Zender.pdf

ACKNOWLEDGMENTS

The authors want to thank Lex Jansen for continuing to publish SAS Proceedings from 1976 to present. The website searches 39,846 papers (as of the date of writing this paper) and is a wealth of information for SAS Users. <https://www.lexjansen.com>. We would also like to thank Jane Eslinger for answering all of our questions on PROC REPORT.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Lisa Mendez
Army MEDCOM
sasebmendez@gmail.com

Richann Jean Watson
DataRich Consulting
richann.watson@datarichconsulting.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brands and product names are trademarks of their respective companies.